
' -----[ Title ]-----------------------------------------------------------------------

'Robotics with the Boe-Bot - IrRoamingAndEscapeCornersAndPhotoAndTempStore.bs2

'Modification of FastIrRoaming (added pulse command).

'

' -----[ Notes ]-----------------------------------------------------------------------

'

'Used rechargeable 1.25V batteries.

'

'The program change from FastIrRoamingAndPulsing.bs2 to the one now shown and is a

'significant improvement. The second problem is now solved.

'

'1st. The Boe-Bot has flipped over in the past. It gets beside at wall

'(angled into it) AND a wheel will drive up the side of a wall.

'

'2nd. The Boe-Bot cannot get out of corners.

'

'3rd. It keeps reseting itself. It keeps doing this because there is not enough

'energy TO power the Red LEDs, servos, emitters, and detectors. Solved this

'problem by turning off the Red LEDs.

' -----[ Program ]---------------------------------------------------------------------

'{$STAMP BS2}                             'Stamp directive.

'{$PBASIC 2.5}                            'PBASIC directive.

DEBUG "Program Running!"

' -----[ Variables/Constants/Pins ]----------------------------------------------------

' Stuff for IRroaming.

IrDetectorLeft    VAR    Bit

IrDetectorRight   VAR    Bit

PulseLeft         VAR    Word

PulseRight        VAR    Word

LoopCount         VAR    Byte

PulseCount        VAR    Byte

FreqDetectable    CON    38500

Piezospeaker      PIN    4                'Speaker

WheelRight        PIN    12               'Wheels left/right.

WheelLeft         PIN    13

IrDetectLeft      PIN    9                'Left emitter/detector pair

IrLedLeft         PIN    8

IrDetectRight     PIN    0                'Right emitter/detector pair

IrLedRight        PIN    2

' Leaving corners

CounterRight      VAR    Nib

CounterLeft       VAR    Nib

' Stuff for data collection

PRRight           PIN    3

TempProbe         PIN    15

Kal               CON    15300

rct               VAR    Word             ' A word variable.

TK                VAR    Word             ' Kelvin temperature.

TC                VAR    Word             ' Degrees Celsius.

TimeRight         VAR    Word

GetSensors        VAR    Word

counter           VAR    Byte

CurLoc            VAR    Word             'EEPROM location

' -----[ Initialization ]--------------------------------------------------------------

FREQOUT Piezospeaker, 2000, FreqDetectable           'Signal program start/reset.

CounterLeft = 1

CounterRight = 1

GetSensors = 0

CurLoc = 0

' -----[ Main Routine ]----------------------------------------------------------------

DO 'Main loop

  GOSUB IrSensing

  GOSUB Movements

  '----- Get out of corners -------------------

  IF (CounterLeft > 4) AND (CounterRight > 4) THEN

    GOSUB Turn_Around

    CounterLeft = 1

    CounterRight = 1

  ENDIF '(CounterLeft > 4) AND (CounterRight > 4)

  '----- Sensors ----------------------------

  GetSensors = GetSensors + 1

  IF (GetSensors = 300) OR (GetSensors = 600) OR (GetSensors = 900) THEN

    GOSUB DataLogging

  ELSEIF (GetSensors = 902) THEN

    GOSUB Ending

  ENDIF

LOOP 'Main loop

Ending:

  END

'------[ Subroutines ]-----------------------------------------------------------------

'----- IR Sensing ---------------------------

IrSensing:

  FREQOUT IrLedLeft, 1, FreqDetectable    'Store IR detection values.

  IrDetectorLeft = IN9

  FREQOUT IrLedRight, 1, FreqDetectable

  IrDetectorRight = IN0

RETURN

'----- Movements ----------------------------

Movements:

  IF (IrDetectorLeft = 0 ) AND (IrDetectorRight = 0) THEN

    PulseCount = 15

    PulseLeft = 650                       'Back up.

    PulseRight = 850

  ELSEIF (IrDetectorLeft = 0) THEN        'Left LED detected.

    CounterLeft = CounterLeft + 1

    PulseCount = 10

    PulseLeft = 850                       'Turn Right.

    PulseRight = 850

  ELSEIF (IrDetectorRight = 0) THEN       'Right LED detected.

    CounterRight = counterRight + 1

    PulseCount = 10

    PulseLeft = 650                       'Turn Left.

    PulseRight = 650

  ELSE                                    'Go forward.

    PulseCount = 1

    PulseLeft = 850                       'No IR detected, go forward.

    PulseRight = 650

  ENDIF '(IrDetectorLeft = 0 ) AND (IrDetectorRight = 0)

  FOR LoopCount = 1 TO PulseCount

    PULSOUT WheelLeft,  PulseLeft

    PULSOUT WheelRight, PulseRight

    PAUSE 15

  NEXT 'LoopCount = 1 TO PulseCount

RETURN

'----- Escape Corners -------------------

Turn_Around:

  GOSUB Back_Up

  FREQOUT Piezospeaker, 2000, 1000        ' Angerey noise.

  GOSUB Turn_Left

  GOSUB Turn_Left

RETURN 'IF (counter > 4)

Back_Up:                                  ' Back up.

  FOR PulseCount = 0 TO 40

    PULSOUT WheelLeft,  650

    PULSOUT WheelRight, 850

    PAUSE 20

  NEXT

RETURN 'Turn_Around:

Turn_Left:                                ' Left turn, about 90-degrees.

  FOR PulseCount = 0 TO 22

    PULSOUT WheelLeft,  650

    PULSOUT WheelRight, 650

    PAUSE 20

  NEXT

RETURN 'Turn_Around:

'----- Data Logging ---------------------

DataLogging:

  DEBUG CR, "Start Data Loging", CR        ' Used to figure out duration of data logging

  FREQOUT Piezospeaker, 500, FreqDetectable

  FOR counter = 1 TO 30                         ' Main routine.

    GOSUB PhotoReadings

    GOSUB TempReadings

    GOSUB SaveReadings

    PAUSE 960

  NEXT

  DEBUG "Stop Data Loging", CR             ' Used to figure out duration of data logging

RETURN

'----- Photo Data ------------------------

PhotoReadings:

  HIGH PRRight                               ' Right RC time measurement.

  PAUSE 2

  RCTIME PRRight, 1, TimeRight

RETURN

'----- Temperature Data ------------------

TempReadings:

  LOW TempProbe               ' Discharge the capacitor.

  RCTIME TempProbe, 0, rct    ' Time for the volts to rise to 1.3 V.

  TK = Kal/rct*10 + (Kal//rct*10/rct) ' Calculate Kelvin

  TC = TK - 273                       ' and Celsius.

RETURN

'----- Log Data --------------------------

SaveReadings:

  CurLoc = CurLoc + 1                        'Increment EEPROM storage location

  WRITE CurLoc, (GetSensors/300)             'Store location

  CurLoc = CurLoc + 1                        'Increment EEPROM storage location

  WRITE CurLoc, TimeRight                    'Store light readings

  CurLoc = CurLoc + 1                        'Increment EEPROM storage location

  WRITE CurLoc, TC                           'Store temperature readings

RETURN


